This *Mathematica* Notebook explores various aspects of disparity energy models as presented in our papers:

Ohzawa, I., DeAngelis, G.C., and Freeman, R.D. Stereoscopic depth discrimination in the visual cortex: Neurons ideally suited as disparity detectors. *Science* **249**: 1037-1041, 1990.

Ohzawa, I., DeAngelis, G.C., and Freeman, R.D. Encoding of binocular disparity by complex cells in the cat's visual cortex. *J. Neurophysiol.* **77**: 2879-2909, 1997.

izumi@pinoko.berkeley.edu

Odd-symmetric Gabor filter

```
Plot[ Exp[-5.5 x^2] Sin[2 Pi x], {x, -1, 1}, PlotPoints -> 40]
```




```
-Graphics-
```

Even-symmetric Gabor filter

```
Plot[ Exp[-5.5 x^2] Cos[2 Pi x], {x, -1, 1}, PlotPoints -> 40]
```



-Graphics-

Squared output of odd-symmetric Gabor filter

```
Plot[ (Exp[-5.5 x^2] Sin[2 Pi x])^2, {x, -1, 1},
PlotRange -> {0.0, 1.0}, PlotPoints -> 40]
```



-Graphics-

Squared output of even-symmetric Gabor filter

```
Plot[ (Exp[-5.5 x^2] Cos[2 Pi x])^2, {x, -1, 1},
PlotRange -> {0.0, 1.0}, PlotPoints -> 40]
```




```
-Graphics-
```

Sum of Squares -> ENERGY

```
Plot[(Exp[-5.5 x^2] Sin[2 Pi x])^2
   + (Exp[-5.5 x^2] Cos[2 Pi x])^2,
{x, -1, 1}, PlotRange -> {0.0, 1.0}, PlotPoints -> 40]
```




```
-Graphics-
```

If sum of absolute values is taken instead of sum of squares, ripples will be apparent in the output profile.

```
Plot[ Abs[Exp[-5.5 x^2] Sin[2 Pi x]]
    + Abs[Exp[-5.5 x^2] Cos[2 Pi x]],
{x, -1, 1}, PlotRange -> {0.0, 1.5}, PlotPoints -> 40]
```



-Graphics-

Complex cell model's binocular response profile [Ohzawa et al. Science 249:1037-1041, 1990]

```
DensityPlot[ 4 -
   (( Exp[-5.5 xL^2] Cos[2 Pi xL]
    + Exp[-5.5 (xR)^2] Cos[2 Pi (xR)] )^2
  + ( Exp[-5.5 xL^2] Sin[2 Pi xL]
    + Exp[-5.5 (xR)^2] Sin[2 Pi (xR)] )^2),
   {xL, -1, 1}, {xR, -1, 1}, PlotRange -> {0.0, 4.0},
   PlotPoints -> 40, Mesh -> False, Frame -> False]
```



-DensityGraphics-

Binocular response profile when complex cells are constructed using absolute value non-linearity instead of squaring.
(This is Fig. 8E of Ohzawa, et al., *J. Neurophysiol. 77: 2879-2909*, 1997)

```
DensityPlot[ 2.8 -
   ( Abs[ Exp[-5.5 xL^2] Cos[2 Pi xL]
     + Exp[-5.5 (xR)^2] Cos[2 Pi (xR)] ]
   + Abs[ Exp[-5.5 xL^2] Sin[2 Pi xL]
     + Exp[-5.5 (xR)^2] Sin[2 Pi (xR)] ]),
   {xL, -1, 1}, {xR, -1, 1}, PlotRange -> {0.0, 2.8},
   PlotPoints -> 40, Mesh -> False, Frame -> False ]
```



-DensityGraphics-

This is the response of a disparity sensor tuned to a non-zero disparity. The model is based on the ***double-quadrature*** organization of RF subunits.

```
DensityPlot[ 3.4 -
   (( Exp[-5.5 xL^2] Cos[2 Pi xL] + Exp[-5.5 xR^2] Sin[2 Pi xR] )^2
 + ( Exp[-5.5 xL^2] Sin[2 Pi xL] - Exp[-5.5 xR^2] Cos[2 Pi xR] )^2),
   {xL, -1, 1}, {xR, -1, 1},
    PlotRange -> {0.0, 3.4},
   PlotPoints -> 40, Mesh -> False]
```



-DensityGraphics-

This is the response of one of the subunits (simple cell) that contributes to the final complex cell response.

```
DensityPlot[
Pos[( -Exp[-5.5 xL^2] Cos[2 Pi xL]
     + Exp[-5.5 xR^2] Sin[2 Pi xR] ) ],
   {xL, -1, 1}, {xR, -1, 1},
   PlotPoints -> 40, Mesh -> False, Axes -> None ]
```



-DensityGraphics-

An opponent energy model by S.F. Bowne, S.P. McKee, and C.W. Tyler (ARVO ’90, #1488, p. 303) uses crossed and uncrossed disparity sensers that consist of incongurugous consine-phase L,R receptive fields.

```
Plot[ {Exp[-5.5 (x+0.2)^2] Cos[2 Pi (x+0.2)],
       Exp[-5.5 (x-0.2)^2] Cos[2 Pi (x-0.2)] },
        {x, -1.2, 1.2},
        PlotRange -> {-0.4, 1.0},
        PlotPoints -> 40]
```



-Graphics-

This is the response of an opponent-energy model proposed by S.F. Bowne, S.P. McKee, and C.W. Tyler (ARVO ’90, #1488, p. 303). Disparity selectivity is achieved, but the position independence is not, as shown by the apparent lack of diagonal elongation.

```
DensityPlot[ ( Exp[-5.5 (xL+0.2)^2] Cos[2 Pi (xL+0.2)]
         + Exp[-5.5 (xR-0.2)^2] Cos[2 Pi (xR-0.2)] )^2
   - ( Exp[-5.5 (xL-0.2)^2] Cos[2 Pi (xL-0.2)]
         + Exp[-5.5 (xR+0.2)^2] Cos[2 Pi (xR+0.2)] )^2,
   {xL, -1, 1}, {xR, -1, 1},
   PlotRange -> {-3.5, 3.5},
   PlotPoints -> 40, Mesh -> False]
```



-DensityGraphics-